

# Introduction to Digital Circuit Design Using Verilog

## BLG 222E - Computer Organization

T. Tolga Sarı, Kadir Özlem, Tuğba Pamay, Mücahit Altıntaş  
{sarita, kadir.ozlem, pamay, maltintas}@itu.edu.tr

Istanbul Technical University

February 25, 2022



# Outline

- 1 Introduction
- 2 Syntax Details
- 3 Xilinx Vivado Design Suite Instalation
- 4 Vivado Design Suite Usage

# Hardware Description Language (HDL)

- Describe hardware using code.
- No need to define what every transistor/gate does when coding.
- Simulate logic before building. Running simulations are quite faster than wiring manually.
- Port previously developed module to another project easily.
- Write parametric units. 8-bit multiplier → 32-bit multiplier.
- Synthesize code into gates and layout (with help of a standard cell library)
- Verilog, VHDL, SystemVerilog...

# Field Programmable Gate Array

- After creating the design, it can be ported to the ASIC(application specific integrated circuit) which is a whole different level of design process or, the design can be realized using FPGA(field programmable gate array). Which is quite easier process.
- An FPGA is basically programmable digital circuit and it consists of LUT(look up table) which is basically RAM tied to the multiplexer and programmable routing logic.




Figure 1: Inside of the FPGA.

- One of the mainstream HDL.
- Has syntax similar to C programming language → More easier to adapt. Also it needs less typing compared to other languages.
- Allows hardware designer to describe language both at high levels (if-else, for-while, switch-case) and at low levels of abstractions.
- Allows programmers to create macros and functions.

# General Module Structure

```
1 module <name>(<port list>);  
2     <Declarations>  
3  
4     <Module Items>  
5 endmodule
```

# General Module Structure

- Module Name is basically function name.
- Inside the port list a list of input, inout and output ports which are referenced in other modules.
- Section specifies data objects as registers, memories and wires as well as procedural constructs such as functions and tasks.
- Assignments, logical blocks, initial constructs are given inside the Module Items
- Let's look at an example code for the AND gate.

# Simple AND Gate

```
1 module and_gate(i_1, i_2, o);
2     // Inputs
3     input    wire    i_1;
4     input    wire    i_2;
5
6     // Outputs
7     output   wire    o;
8
9     assign o = i_1 & i_2;
10 endmodule
```




Figure 2: RTL Schematic of AND Gate Module



# Simple AND Gate

- At first, module name and its connections are defined (Line 1).
- Then the connection directions and the data type is defined (Lines 3-7). Here A and B are the inputs of the **and\_gate** and the output will be C.
- **wire** data type is the most basic data type in Verilog. It can be treated as physical wires.
- Lastly these wires are connected using **assign** keyword.

# Using a Module Inside Another Module

- We don't need to re-code frequently recurring code parts in HDL. We can use them anywhere of a module given that the connections are correct.
- This will help us to divide the digital design into smaller and less complex parts.
- Also good design on the paper → less headache when coding the hardware.

# Using a Module Inside Another Module

```
1 module module_name(wire1, wire2);  
2     ...  
3     ...  
4     and_gate G1(.i_1 (some_wire_1),  
5                 .i_2 (some_wire_2),  
6                 .o   (some_wire_3));  
7     ...  
8     ...  
9 endmodule
```

# Simple Sum of Products

- Let's create a digital circuit for  $D = AB + BC$ .
- This can be done by either first creating an OR gate and combining them in another module, or realizing this function in another module.
- Both methods will be demonstrated.

# Simple Sum of Products

```
1 module or_gate(i_1, i_2, o);
2     // Inputs
3     input    wire    i_1;
4     input    wire    i_2;
5
6     // Outputs
7     output   wire     o;
8
9     assign o = i_1 | i_2;
10 endmodule
```




Figure 3: RTL Schematic of OR Gate Module

# Simple Sum of Products

```
1 module SOP(A, B, C, D);
2     // Inputs
3     input    wire    A;
4     input    wire    B;
5     input    wire    C;
6
7     // Outputs
8     output   wire    D;
9
10    // Intermediate Wires
11        wire    AB;
12        wire    BC;
13
14    and_gate AND1(.i_1(A ), .i_2(B ), .o(AB));
15    and_gate AND2(.i_1(B ), .i_2(C ), .o(BC));
16    or_gate  OR1(.i_1(BC), .i_2(AB), .o(D) );
17 endmodule
```

# Simple Sum of Products




Figure 4: RTL Schematic of SOP Module

# Simple Sum of Products

- This can be done by coding the boolean function in one line also:

```
1 module SOP(A, B, C, D);
2     // Inputs
3     input    wire    A;
4     input    wire    B;
5     input    wire    C;
6
7     // Outputs
8     output   wire    D;
9
10    assign D = (A & B) | (B & C);
11 endmodule
```



# Simple Sum of Products




Figure 5: RTL Schematic of Alternative SOP Module

# Simple Sum of Products

- So, we created our module. How can we simulate it?
- We need to create a test bench in which we send input signals to our module.
- The test bench code structure is fairly simple and follows the general code structure.
- We basically create a module that has our **SOP** and applies inputs to this module at specific times.
- The inputs of the uut(unit-under-test) will be data type called **reg** and the output will be **wire** data type. **reg** data type will be discussed later.

# Simple Sum of Products

```
1 module SOP_test();
2     // Test module has no inputs and outputs.
3     reg          a;
4     reg          b;
5     reg          c;
6
7     wire          d;
8     SOP uut(.A(a), .B(b), .C(c), .D(d));
9
10    initial begin
11        // Assign test values and wait 250 time units.
12        a=0; b=0; c=0; #250;
13        a=0; b=1; c=0; #250;
14        a=1; b=0; c=1; #250;
15        a=1; b=1; c=1; #250;
16    end
17 endmodule
```

# Simple Sum of Products Simulation




Figure 6: Simulation Timeline for SOP Test Module

- Let's discuss more detailed syntax of the Verilog.
- We will discuss more indept operators.
- Multi bit operands.
- **reg** data type and initial block.
- Always block

# List of Operators

Verilog Operator	Name
&	Bitwise AND
	Bitwise OR
~	Bitwise NOT
^	Bitwise XOR
+	Arithmetic Addition
-	Arithmetic Subtraction
*	Arithmetic Multiply
/	Arithmetic Division
<<	Logical Left Shift
>>	Logical Right Shift
{ }	Concatenation

Table 1: Commonly used Verilog Operators.

- Until now, we have only dealt with the 1-bit data.
- Using the multi bit data is fairly simple with the Verilog. Only needed thing is specifying the bit length of the data.
- For example let's look at 32-bit AND gate.

# Multi Bit Data

```
1 module and_gate_32(i_1, i_2, o);
2     // Inputs
3     input    wire [31:0] i_1;
4     input    wire [31:0] i_2;
5
6     // Outputs
7     output   wire [31:0] o;
8
9     assign o = i_1 & i_2;
10 endmodule
```




Figure 7: RTL Schematic of 32-bit AND Gate Module



- Numbers in Verilog can be specified in many ways. All of them needs bit count and base that is used. Let's look at some examples.

```
1      ...
2      // foo and bar are 8-bit numbers.
3      // All of the things below mean the same thing.
4      // Which is foo = bar + 153
5      assign foo = bar + 8'b10011001;
6      assign foo = bar + 8'b1001_1001;
7      assign foo = bar + 8'h99;
8      assign foo = bar + 8'd153;
9      ...
```

- Let's finish this section by looking at the testbench of the 32-bit and gate.

# Testbench for 32-bit AND Gate

```
1 module and_gate_32_test();
2     // Test inputs
3     reg [31:0] A;
4     reg [31:0] B;
5     // Test outputs
6     wire [31:0] C;
7
8     and_gate_32 uut(.i_1(A), .i_2(B), .o(C));
9
10    initial begin
11        A = 32'hFFFFFFFF; B = 32'hFF00FF00; #250;
12        A = 32'h00000000; B = 32'hFFFFFFFF; #250;
13        A = 32'h55555555; B = 32'hFF00FF00; #250;
14        A = 32'hFFFFFFFF; B = 32'hFFFF0000; #250;
15    end
16
17 endmodule
```

# 32-bit AND Gate Simulation




Figure 8: Simulation Timeline for 32-bit AND Gate Module

- Aside from logical high and low, the bits in a variable can take more values in the Verilog.
- The bit can be in high impedance mode, which is donated as 32'dZ for 32-bit data.
- The bit value can be undetermined which is donated as 32'dX for 32-bit data.
- That is why even if we define behaviour for all bit combinations we still have to create a default case (we will discuss switch case in the following slides).


- **reg** data type is more capable data type that can be used in more complex code blocks in Verilog.
- It can be used to define flip-flops and other memory elements such as RAM or Cache.
- Unlike its name, using reg data type **will not generate flip-flops always**.
- Its power lies in the **always** code blocks.

- **always** block consists of 2 parts.
- It has a sensitivity window that determines when the updates in the always block will be applied. For example always block can be sensitive to the falling edge of the clock.
- After defining the sensitivity window, functionality of the always block will be coded.
- In always code block, more high level abstractions(if-else, switch-case) can be used.
- **ONLY THE REG DATA TYPE CAN BE UPDATED IN THE ALWAYS BLOCK.**

# Example Counter

```
1      ...
2          reg [31:0] A;
3      ...
4      // Simple counter that counts up by using
5      // ADDER In each positive edge of the clock.
6      always @(posedge clock)
7      begin
8          A <= A + 32'd1;
9      end
10     ...
```

## Example Counter



### Figure 9: RTL Schematic of Example Counter



- Two types of assignments can be done in the always block.
- In the last example we saw parallel( $\leq$ ) assignment. Despite it's name, this assignment **overrides** all of other previous assignments to that variable in that always block.
- The other type of assignment is sequential( $=$ ) assignment. With this one we can cascade multiple operations in one always block.
- **ONLY ONE TYPE OF ASSIGNMENT CAN BE USED IN THE SAME ALWAYS BLOCK.**

# Parallel vs Sequential Assignment

```
1      ...
2          reg [31:0] A;
3          reg [31:0] B;
4      ...
5      always @(posedge clock)
6      begin
7          A <= A + 32'd1; // This line will be
8          A <= A << 2;    // Overridden by this line.
9      end                // A = 4A
10     ...
11     ...
12     always @(posedge clock)
13     begin
14         B = B + 32'd1;
15         B = B << 2;
16     end                // B = 4(B + 1)
17     ...
```

# Parallel vs Sequential Assignment




Figure 10: RTL Schematic of Parallel and Sequential Assignment

# Counter With Reset Using If-Else Block

```
1      ...
2          reg [31:0] A;
3      ...
4      always @(posedge clock or negedge reset)
5      begin
6          if(!reset)                // Reset is active low.
7              begin
8                  A <= 32'd0;
9              end
10             else
11                 begin
12                     A <= A + 32'd1;
13                 end
14         end
15     ...
```

# Counter With Reset Using If-Else Block




Figure 11: RTL Schematic of Counter With Reset Using If-Else Block

# Sample FSM using Switch-Case

```
1      case(state)
2          2'b00:      begin
3                      A <= 32'd1234;
4                      state <= 2'b11;
5                      end
6          2'b01 :      begin
7                      A <= 32'd2222;
8                      state <= 2'b00;
9                      end
10         2'b11 :      begin
11                     A <= 32'd1773;
12                     state <= 2'b01;
13                     end
14         default :      begin
15                     A <= 32'd0000;
16                     state <= 2'b00;
17                     end          // State can be
18     endcase              // 2'bXX or 2'bZZ
```

## Sample FSM using Switch-Case




Figure 12: RTL Schematic of Sample FSM using Switch-Case

# Combinational Always Blocks

```
1      ...
2          reg [3:0] A;
3          reg [1:0] B;
4      // Simple 2-to-4 Decoder.
5      // * means when any variable in always block
6      // Changes its value apply always block
7      always @(*)
8      begin
9          case(B)
10             2'b00:    A <= 4'b0001;
11             2'b01:    A <= 4'b0010;
12             2'b10:    A <= 4'b0100;
13             2'b11:    A <= 4'b1000;
14
15             default: A <= 4'b0000;
16          endcase
17      end
18      ...
```



# Combinational Always Blocks




Figure 13: RTL Schematic of Combinational Always Blocks

# General Rules of Thumb

- Never assign to same variable in different always blocks. Race condition between them almost always create problems.
- Try to separate your FSM designs into 2 parts. In one of only update the state values using **always@(posedge clock)** and in the other calculate rest of the things. Doing this separation correctly will cause less problems when coding.
- You are coding hardware, never forget that. So do not try to do many things in one clock cycle.
- Never use initial blocks in your design files. It is impossible to physically set a bit when module powers up. Set your registers using reset functionality instead.

# Example Properly Coded FSM

```
1 module coded_fsm(clock, reset, o);
2     input    wire        clock;
3     input    wire        reset;
4         reg    [1:0]    state_reg_q;
5         reg    [1:0]    state_reg_d;
6     output   reg    [7:0]    o;
7     always @(posedge clock or negedge reset) begin
8         if(!reset)
9             begin
10                 state_reg_q <= 2'd0;
11                 o <= 8'd0;
12             end
13         else
14             begin
15                 state_reg_q <= state_reg_d;
16                 o <= o + state_reg_q;
17             end
18     end
```

# Example Properly Coded FSM

```
19  always @(*)
20  begin
21      case(state_reg_q)
22          2'b00:    state_reg_d <= 2'b10;
23          2'b01:    state_reg_d <= 2'b00;
24          2'b10:    state_reg_d <= 2'b01;
25
26          default:  state_reg_d <= 2'b00;
27      endcase
28  end
29  endmodule
```

# Example Properly Coded FSM




Figure 14: RTL Schematic of Example Properly Coded FSM

# Creating Clock in the Testbench

```
1 module coded_fsm_tb();
2     reg          clock;
3     reg          reset;
4     wire [7:0]   out;
5
6     coded_fsm module_test(clock, reset, out);
7     initial begin
8         clock=0; reset=1; #100; //Initiate signals
9         reset=0; #30;
10        reset=1; #200;
11        reset=0; #20;
12        reset=1; #150;
13        $finish; //Terminate simulation
14    end
15    always begin
16        clock = ~clock; #25; //Toggle clock signal
17    end
18 endmodule
```

# Simulation for Coded FSM




Figure 15: Simulation Timeline for Coded FSM

# Example ALU Using Combinational Always Block

```
1  always @(*)
2  begin // Example 8-bit ALU
3      case(operation)
4          // Addition
5          2'b00:    o <= A + B;
6          // Subtraction
7          2'b01:    o <= A - B;
8          // Circular left shift
9          2'b10:    o <= {A[6:0],A[7]};
10         // Circular right shift
11         2'b11:    o <= {A[0],A[7:1]};
12
13         default:  o <= 8'd0;
14     endcase
15 end
```



# Example ALU Using Combinational Always Block




Figure 16: RTL Schematic of ALU Module

# Parametric Modules

- We will generate n-bit ripple carry adder/subtractor.
- First, we need to code a full adder.

```
1 module FA (a, b, cin, s, cout);
2     input    wire    a;
3     input    wire    b;
4     input    wire    cin;
5     output   wire    s;
6     output   wire    cout;
7             wire    a_xor_b;
8
9     assign a_xor_b = a ^ b;
10    assign s = a_xor_b ^ cin;
11    assign cout = (a_xor_b & cin) | (a & b);
12 endmodule
```

# FSM Schematic




Figure 17: FSM schematic that is generated.

- After coding full adder. We will use these full adders as basic building blocks for the ripple carry adder.
- n-bit ripple carry adder needs n full adders. To parametrize this using Verilog, we will use **generate** code block.
- Basically inside generate block, we will create a iterator (called **genvar**) and use this iterator in a for loop to create our parametrized ripple carry adder.

# n-bit Ripple Carry Adder

```
1 module RPA_nbit #(parameter n = 4)
2     (a, b, select, s, cout);
3     input  wire    [n-1:0]    a;
4     input  wire    [n-1:0]    b;
5     // Addition or Subtraction ?
6     input  wire                                select;
7
8     output wire    [n-1:0]    s;
9     output wire                                cout;
10
11    // Will connect carry ports of FAs
12        wire    [n :0]        miniCout;
13        wire    [n-1:0]        c;
14    // If select=1 in then subtraction happens.
```

# n-bit Ripple Carry Adder

```
14 // If select=1 in then subtraction happens.
15 assign c = (select) ? ~b : b;
16 assign miniCout[0] = select;
17 assign cout = miniCout[n];
18
19 generate
20     genvar x;
21     for(x = 0; x < n; x = x + 1) begin: RPAAnbit
22         FA RPA( .a(a[x]),
23                 .b(c[x]),
24                 .s(s[x]),
25                 .cin(miniCout[x]),
26                 .cout(miniCout[x+1]));
27     end
28 endgenerate
29 endmodule
```

# n-bit Ripple Carry Adder Schematic




Figure 18: 4-bit ripple carry adder.

# n-bit Ripple Carry Adder Testbench

```
1 module Test_RPA_nbit;
2     parameter N = 8; // Since we parametrized RPA
3                       // Its width can be changed.
4     // Inputs
5     reg [N-1:0] a;
6     reg [N-1:0] b;
7     reg select;
8     // Outputs
9     wire [N-1:0] s;
10    wire cout;
11    // #(.n(N)) overrides the n parameter of
12    // module to 8 making it 8 bit adder.
13    RPA_nbit #(.n(N)) uut(.a(a),
14                          .b(b),
15                          .select(select),
16                          .s(s),
17                          .cout(cout));
```



# n-bit Ripple Carry Adder Testbench

```
18     initial begin
19         // Initialize Inputs
20         a = 8'h11; b = 8'h11; select = 0; #100;
21         a = 8'h31; b = 8'hff; select = 0; #100;
22         a = 8'h53; b = 8'hac; select = 0; #100;
23         a = 8'h45; b = 8'hbc; select = 0; #100;
24         a = 8'h84; b = 8'h22; select = 0; #100;
25
26         a = 8'h01; b = 8'h0a; select = 1; #100;
27         a = 8'h0a; b = 8'h0b; select = 1; #100;
28         a = 8'h0c; b = 8'h0f; select = 1; #100;
29         a = 8'hff; b = 8'h0f; select = 1; #100;
30         a = 8'h0f; b = 8'h0a; select = 1; #100;
31     end
32 endmodule
```



- Xilinx is the mainstream FPGA company. Its products are widely used in the industry.
- To program their FPGAs Xilinx first developed ISE Design Platform.
- After version 14.7, Xilinx have created another design platform called Vivado. Which is more versatile and more inline with the industry standards.
- Both of these design platforms have built-in simulators.
- After synthesizing(mapping them to the cells) the design's RC values can be extracted and simulations with propagation delays can be done.

# Install Xilinx Vivado (WebPack version)

[Products](#) [Company](#) **AMD**  
**XILINX**

## Vivado Design Suite - HLx Editions - 2017.4 Full Product Installation

**Important**

We strongly recommend to use the web installers as it reduces download time and saves significant disk space.

Please see [Installer Information](#) for details.

Vivado HLx 2017.4: WebPACK and Editions - Windows Self Extracting Web Installer (EXE - 51.3 MB)	Download Includes	Vivado Design Suite HLx Editions (All Editions)
MD5 SUM Value : 39677d35779915411487e5d89fe27ee8	Download Type	Full Product Installation
Vivado HLx 2017.4: WebPACK and Editions - Linux Self Extracting Web Installer (BIN - 100.7 MB)	Last Updated	Dec 20, 2017
MD5 SUM Value : 915928e0f33af22f1370acc6ac32c2b5	Answers	<a href="#">2017.x - Vivado Known Issues</a>
Vivado HLx 2017.4: All OS installer Single-File Download (TAR/GZIP - 16.17 GB)	Documentation	<a href="#">2017.4 - Release Notes</a>
MD5 SUM Value : 5a1bb63f893f0039e34c9f412dd063a9	Enablement	<a href="#">License Solution Center</a>

**Figure 20:** Download proper *Xilinx Unified Installer 2017.4* executable file with your operating system from <https://www.xilinx.com/support/download.html>

# Installation of Xilinx Vivado on Linux

- To install Xilinx Vivado on linux, you can watch the video on url:  
<https://www.youtube.com/watch?v=cQHe651AEaM>.

# Install Xilinx Vivado (WebPack version)




Figure 21: Welcome page. After executable file is double clicked, this page appears.

# Install Xilinx Vivado (Webpack version)

Vivado 2017.4 Installer - Select Install Type

Select Install type

Please select install type and provide your Xilinx.com user ID and password for authentication.

User Authentication

Please provide your Xilinx user account credentials to download the required files.  
If you don't have an account, [please create one](#). If you forgot your password, you can [reset it here](#).

User ID

Password

☒ Download and Install Now

Select your desired device and tool installation options and the installer will download and install just what is required. Downloaded installation files will be saved for future use. NOTE: Future installs using these downloaded files will be restricted to the options selected during this install. For access to all options later, choose "Download Full Image".

☐ Download Full Image (Install Separately)


The installer will download an image containing all devices and tool options for later installation. Use this option if you wish to install a full image on a network drive or allow different users maximum flexibility when installing.

Copyright © 1986-2022 Xilinx, Inc. All rights reserved.

< Back Next > Cancel

Figure 22: You should enter your Xilinx account information here. If you don't have any, you must sign up firstly.

# Install Xilinx Vivado (WebPack version)



Vivado 2017.4 Installer - Accept License Agreements

**Accept License Agreements**

Please read the following terms and conditions and indicate that you agree by checking the I Agree checkboxes.

**Xilinx Inc. End User License Agreement**

By checking "I AGREE" below, or OTHERWISE ACCESSING, DOWNLOADING, INSTALLING or USING THE SOFTWARE, YOU AGREE on behalf of licensee to be bound by the agreement, which can be viewed by [clicking here](#).

☒ I Agree

**WebTalk Terms And Conditions**

By checking "I AGREE" below, I also confirm that I have read [Section 13 of the terms and conditions](#) above concerning WebTalk and have been afforded the opportunity to read the WebTalk FAQ posted at <https://www.xilinx.com/products/design-tools/webtalk.html>. I understand that I am able to disable WebTalk later if certain criteria described in Section 13(c) apply. If they don't apply, I can disable WebTalk by uninstalling the Software or using the Software on a machine not connected to the internet. If I fail to satisfy the applicable criteria or if I fail to take the applicable steps to prevent such transmission of information, I agree to allow Xilinx to collect the information described in Section 13(a) for the purposes described in Section 13(b).

☒ I Agree

**Third Party Software End User License Agreement**

By checking "I AGREE" below, or OTHERWISE ACCESSING, DOWNLOADING, INSTALLING or USING THE SOFTWARE, YOU AGREE on behalf of licensee to be bound by the agreement, which can be viewed by [clicking here](#).

☒ I Agree

Copyright © 1986-2022 Xilinx, Inc. All rights reserved.

< Back Next > Cancel

Figure 23: After all agreement check boxes are clicked, please click "Next".



# Install Xilinx Vivado (WebPack version)




Figure 24: At this step, you must select "Vivado HL WebPACK" option.

# Install Xilinx Vivado (WebPACK version)






Figure 25: By selecting minimum configuration, you can decrease download and disc space requirements. Then, you can directly go to the next page.

# Install Xilinx Vivado (WebPack version)



**Figure 26:** Be careful about "Disk Space Required" information. According to these information, you can easily change your installation directory in "Installation Options" section.

# Install Xilinx Vivado (WebPACK version)



**Figure 27:** Your final configurations are listed here. Then, click "Install" button to start installation.

# Install Xilinx Vivado (WebPack version)




Figure 28: Installation just started.

# Using Vivado




Figure 29: Open Vivado, and create a new project.

# Using Vivado




Figure 30: Click next.

# Using Vivado

**New Project**

**Project Name**  
Enter a name for your project and specify a directory where the project data files will be stored.

Project name:

Project location:

☒ Create project subdirectory

Project will be created at: /home/tolga/blg242/and\_gate

Figure 31: Enter project name.






Figure 32: The project will be an RTL project.

# Using Vivado

**New Project**

**Default Part**  
Choose a default Xilinx part or board for your project.

**Parts** | Boards

[Reset All Filters](#)

Category: All Package: All Temperature: All  
Family: All Speed: All Static power: All

Search: xc7a100tcsg (4 matches)

Part	I/O Pin Count	Available IOBs	LUT Elements	FlipFlops	Block RAMs	Ultra RAMs	DSPs
xc7a100tcsg324-3	324	210	63400	126800	135	0	240
xc7a100tcsg324-2	324	210	63400	126800	135	0	240
xc7a100tcsg324-2L	324	210	63400	126800	135	0	240
xc7a100tcsg324-1	324	210	63400	126800	135	0	240

< Back Next > Finish Cancel

Figure 33: Select part as xc7a100tcsg324-1

# Using Vivado




Figure 34: Click finish.




Figure 35: Add new sources.

# Using Vivado




Figure 36: At first we will add design sources.

# Using Vivado




Figure 37: You can add existing files or create a brand new file.

# Using Vivado





Figure 38: File type will be Verilog.



**Figure 39:** When creating a file Vivado gives us option to specify inputs and outputs before creation.



# Using Vivado





Figure 40: Code your module.



**Figure 41:** Now, we are ready to simulate. We need to create simulation source which will be the testbench.

# Using Vivado




Figure 42: Simulation file will be Verilog file.




Figure 43: Remember testbench has no I/O.

# Using Vivado




Figure 44: Code your testbench.




Figure 45: We are now ready to simulate. Click Run simulation and select Behavioral Simulation.

# Using Vivado





Figure 46: A waveform will be opened. You can see your applied inputs.

# Using Vivado



**Figure 47:** To make following waveform easier you can change the radix(base) of the numbers displayed in the waveform.



# Using Vivado




Figure 48: Waveform numbers are now decimal.




Figure 49: Open elaborated design to see system schematic.




Figure 50: The system schematic.